

Σχολή Ηλεκτρολόγων Μηχανικών και Μηχανικών Υπολογιστών
Ακαδημαϊκό Έτος 2020-2021



Εξάμηνο 7ο

Εργαστήριο Μικροϋπολογιστών

1η Εργαστηριακή Αναφορά

Κωστάκη Χριστίνα (el18136)
Τσάφος Αλέξανδρος (el18211)

1η Άσκηση

```
IN 10H

LXI B,03E8H      ;1s delay
MVI D,00H        ;clock counter

START:
LDA 2000H        ;input
PUSH PSW
ANI 0FH          ;4LSB
MOV E,A          ;storing input in E
POP PSW
ANI 80H          ;MSB check
CPI 80H          ;if MSB is set, start counting up
JZ COUNTUP       ;else continue checking
JMP START

COUNTUP:
LDA 2000H        ;MSB check
ANI 80H
CPI 80H
JNZ STOPUP
CALL DELB        ;delay 1s
MOV A,D
CMP E            ;compare counter D to limit E
JZ COUNTDOWN     ;if equal, start counting down
CMA              ;LEDs implemented with reverse logic
STA 3000H        ;output
INR D            ;counter increment and repeat
JMP COUNTUP

COUNTDOWN:
LDA 2000H        ;MSB check
ANI 80H
CPI 80H
JNZ STOPDOWN
CALL DELB
MOV A,D
CPI 00H          ;compare counter D to zero
JZ START         ;if equal, check for new input
CMA
STA 3000H
DCR D            ;counter decrement and repeat
JMP COUNTDOWN
```

```

STOPUP:                ;loop for MSB check
LDA 2000H
ANI 80H
CPI 80H
JNZ STOPUP
JMP COUNTUP

```

```

STOPDOWN:              ;loop for MSB check
LDA 2000H
ANI 80H
CPI 80H
JNZ STOPDOWN
JMP COUNTDOWN

```

```

END

```

2η Άσκηση

```

IN 10H

```

```

START:
MVI A,10H    ;screen switch-off
STA 0900H
STA 0901H
STA 0902H
STA 0903H
STA 0904H
STA 0905H

CALL KIND    ;reading X
RLC
RLC
RLC
RLC          ;mul x16
MOV B,A      ;storing decades in B
CALL KIND    ;reading Y
ADD B        ;A stores the number in hex
CALL DECA
LXI D,0900H  ;initial address for STDM
CALL STDM    ;printing XY in dec on display
CALL DCD
JMP START

```

```

DECA:        ;converting data of A hex->dec and storing it to memory

```

```

PUSH H
PUSH D
MVI D,FFH ;initiating counter

HUN: ;hundred's counter
INR D
SUI 64H ;XY - 100
JNC HUN ;while (XY-100)>0, increase D
ADI 64H ;re-adding to get decades
LXI H,0902H ;storing hundreds in the address of the 3rd LSB
MOV M,D
MVI D,FFH ;re-initiating counter

DEC: ;decade's counter
INR D
SUI 0AH ;XY - 10
JNC DEC ;while (XY-10)>0, increase D
ADI 0AH ;re-adding to get units
LXI H,0901H ;storing decades in the address of the 2nd LSB
MOV M,D
STA 0900H ;storing units in the address of the 1st LSB
POP D
POP H
RET

END

```

3η Άσκηση

```

IN 10H

LXI B,01F4H ;0.5s Delay
MVI D,01H ;train at LSB
MOV A,D ;counter D
CMA ;reverse logic of LEDs
STA 3000H

START:
LDA 2000H ;check for MSB
MOV E,A
ANI 80H
CPI 80H
JNZ START ;if MSB is set, check LSB for direction
MOV A,E
ANI 01H

```

```

CPI 01H
JNZ START ;if LSB is unset, move left
JMP STARTL

STARTL:
CALL DELB ;extra 0.5s delay at LSB
GOLEFT:
LDA 2000H ;check for MSB
MOV E,A
ANI 80H
CPI 80H
JNZ GOLEFT
MOV A,E
ANI 01H ;check for LSB
CPI 01H
JNZ GOR
MOV A,D
CPI 80H
JZ STARTR ;if the train reaches the MSB, change direction
RLC ;else keep moving left
MOV D,A
CMA
STA 3000H ;LED on
CALL DELB
JMP GOLEFT

STARTR:
CALL DELB ;extra 0.5s delay at MSB
GORIGHT:
LDA 2000H ;check for MSB
MOV E,A
ANI 80H
CPI 80H
JNZ GORIGHT
MOV A,E
ANI 01H ;check for LSB
CPI 01H
JNZ GOL
MOV A,D
CPI 01H
JZ STARTL ;if the train reaches the LSB, change direction
RRC ;else keep moving right
MOV D,A
CMA
STA 3000H ;LED on
CALL DELB

```

JMP GORIGHT

GOL: ;loop for LSB check

LDA 2000H

ANI 01H

CPI 01H

JZ STARTL ;if LSB is set again, move left

JMP GOL

GOR: ;loop for LSB check

LDA 2000H

ANI 01H

CPI 01H

JZ STARTR ;if LSB is set again, move right

JMP GOR

END